
At the conclusion of this unit, you will be able to:

Events: Unit Objectives

Events: Overview

Car

Vehicle registration office

Car rental company

* main program

CREATE OBJECT r_car EXPORTING 

...

Sender

Receiver / handler

... 

Event:  

“vehicle_created

”


· By triggering an event, an object or class announces a change of state, or that a certain state has been achieved.

· In the example, the class car triggers the event created. Other classes subscribe to this event (triggered when a car is instantiated) and process it. The car rental company wants to be informed of completion; the car is registered at the vehicle registration office. 

· Note:
The events discussed here are not the events of the ABAP runtime system (such as INITIALIZATION,
START-OF-SELECTION, and so on) and not the events of background processing or workflow. 

[image: image4.wmf]ã

SAP

AG 2002

Features

l

Looser linkage than for a method call

n

Triggering object makes status change known

n

Handler objects can show interest and react

l

Different communication model

n

The trigger does not know the user; 

user's reaction is of no interest to trigger

l

Areas of use

n

GUI implementations

n

Conformity with other object models: COM, ActiveX controls, 

OpenDoc


· Events link objects or classes more loosely than direct method calls do. Method calls establish precisely when and in which statement sequence the method is called. However, with events, the reaction of the object to the event is triggered by the event itself.

· Events are most often used in GUI implementations.

· Other external object models, such as COM, ActiveX controls, and so on, also provide events.

Triggering and Handling Events: Overview

l

Triggering events

§

Class defines event

(

EVENTS, CLASS

-

EVENTS

)

§

Object or class triggers event 

(

RAISE EVENT

)

l

Handling events

§

Event handler class defines and implements event 

handler method 

([CLASS

-

]METHODS...

FOR EVENT ...

OF

...

)

§

Event handler object or handler class registers itself to 

specific events at runtime

(

SET HANDLER

)

Rules:

1

2

3

4


· At the moment of implementation, a class defines its:

· Instance events (using the EVENTS statement)

· Static events (using the CLASS-EVENTS statement)

· Classes or their instances that receive a message when an event is triggered at runtime and want to react to this event define event handler methods.
Statement:
[CLASS-]METHODS <handler_method> FOR EVENT <event> OF <classname>.
· These classes or their instances are registered to one or more events at runtime.
Statement:
SET HANDLER <handler_method> FOR <reference>. (for instance events)
SET HANDLER <handler_method>. (for static events)

· A class or instance can trigger an event at runtime using the RAISE EVENT statement.

Defining and Triggering Events: Syntax

CLASS <

classname

> DEFINITION.

EVENTS: <

event

> EXPORTING VALUE(<ex_par>) TYPE <type>.

CLASS 

lcl

_vehicle DEFINITION.

PUBLIC SECTION.

METHODS constructor IMPORTING 

...

.

EVENTS 

vehicle_created.

...

ENDCLASS.

CLASS <

classname

> IMPLEMENTATION.

METHOD  <m>.

RAISE EVENT <

event

> EXPORTING <ex_par> = <

act

_par>.

CLASS lcl_vehicle IMPLEMENTATION.

METHOD constructor.

...

RAISE EVENT

vehicle_created.

ENDMETHOD.

ENDCLASS.

car

1

2

“vehicle_created”


· Both instance and static events can be triggered in instance methods.

· Only static events can be triggered in static methods.

· Events can only have EXPORTING parameters which must be passed by value.

· Triggering an event using the statement RAISE EVENT has the following effect:

· The program flow is interrupted at that point

· The event handler methods registered to this event are called and processed

· Once all event handler methods have been executed, the program flow continues

· If an event handler method in turn triggers an event, then the program flow is again interrupted and all event handler methods are executed (nesting).

Handling and Registering Events

registration

car1

truck

carrier

Handler

“vehicle_created

· Events are registered using the SET HANDLER statement. Registration is only active at program runtime. Events cannot be persistent.

· You want to register an object to an event belonging to another object. The SET HANDLER statement enters the registration in that object's list. All handlers for one event are entered in this list.

· When the event is triggered, the list shows which event handler methods need to be called.

Event Handler Methods

CLASS <

class

_handle> DEFINITION.

METHODS: <

on

_

event

> FOR EVENT <

event

>

OF <

classname

> | <

interface

>

IMPORTING <ex_par1> ... <ex_

parN

> [sender].

CLASS 

lcl

_rental DEFINITION.

...

PRIVATE SECTION.    

METHODS: add_vehicle FOR EVENT vehicle_created OF 

lcl

_vehicle

IMPORTING 

sender

.

ENDCLASS.

car1

rental

add_vehicle

3

“vehicle_created”


· Event handler methods are triggered by events (RAISE EVENT), although they can also be called like normal methods (CALL METHOD). 

· The interface of the event handler method consists solely of IMPORTING parameters. You can only use parameters from the definition of the corresponding events (event interface). An event interface, which only has EXPORTING parameters, is defined using the EVENTS statement in the declaration of the event. The parameters are typed in the event definition and the typing is passed to the event handler method, that is, the interface parameters of the event handler method cannot be typed in the definition of the event handler method.  
In addition to the explicitly defined event interface parameters, the implicit parameter sender can also be listed as an IMPORTING parameter for instance events. This passes on a reference to the object that triggered the event.                                                                                    

CLASS

lcl

_rental IMPLEMENTATION.

METHOD constructor.

...

SET HANDLER

add_vehicle FOR ALL INSTANCES.

ENDMETHOD.

ENDCLASS.

Registering for an Event: Syntax

SET HANDLER <

ref

_handle>

-

><on_event> 

FOR <

ref

_sender> | FOR ALL INSTANCES

[ACTIVATION <

var

>].

CLASS

lcl

_rental DEFINITION.

...

PRIVATE SECTION.

METHODS: add_vehicle FOR EVENT vehicle_created OF 

...

ENDCLASS.

car1

rental

set handler 

4

“vehicle_created

”


· When an event is triggered, only those event handler methods are executed that have, by this point, registered themselves using SET HANDLER.

· You can register an event using ACTIVATION 'X', and deregister it using ACTIVATION space. If you do not specify ACTIVATION, then the event registers (default behavior).

· You can register several methods in one SET HANDLER statement: 
SET HANDLER <ref_handle1>-><handler_method1>
            ... 
            <ref_handleN>-><handler_methodN>
            FOR <ref_sender> | FOR ALL INSTANCES.

Registered

object

Handler

method

· Every object that has defined events has an internal table, the handler table. All objects that have registered for events are entered in this table together with their event handler methods.

· Objects that have registered themselves for an event that is still "active" also remain "active". The methods of these objects are called when the event is triggered, even if they can no longer be reached using main memory references.

Event Handling: Features

l

Event handling is sequential

l

Sequence in which event handler methods are called is not define

d

l

With regard to the Garbage Collector, registration has the same 

effect 

as a reference to the registered object

n

Registered objects are never deleted

l

The visibility of an event defines authorization for event handl

ing 

l

The visibility in an event handler method defines authorization 

for 

using SET HANDLER statements

n

Event handler methods, however, can only have the same visibilit

y or more 

restricted visibility than the events they refer to


· If several objects have registered for an event, then the sequence in which the event handler methods are called is not defined, that is, there is no guaranteed sequence in which the event handler methods are called.

· If a new event handler is registered in an event handler method for an event that has just been triggered, then this event handler is added to the end of the sequence and is then also executed when its turn comes.
If an existing event handler is deregistered in an event handler method, then this handler is deleted from the event handler method sequence.

· Events are also subject to the visibility concept and can therefore be either public, protected, or private. Visibility specifies who can handle an event:

· PUBLIC:
All users

· PROTECTED:
Only users within that class or its subclasses

· PRIVATE:
Only users within that class

· Event handler methods also have visibility attributes. Event handler methods, however, can only have the same visibility or more restricted visibility than the events they refer to. The visibility of event handler methods establishes authorization for SET HANDLER statements; SET HANDLER statements can be used: Everywhere, in the class and its subclasses, or only within the class.

You are now able to:

Events: Unit Summary


 Events 

















Exercises

	Unit: Events

Topic: Triggering and Handling Events

	[image: image14.png]



	At the conclusion of these exercises, you will be able to:

· Define and trigger events

· Handle events

· Register event handler methods

	As soon as a new airplane is created, this event must be made known to the airline. .

	Model solution:
SAPBC401_EVES_MAIN_A

SAPBC401_VEHD_H include program

SAPBC401_EVES_A include program


1
The add_airplane method of the class lcl_carrier will no longer be explicitly called in the main program, but be triggered automatically from the class lcl_airplane.
Triggering an event when a plane is created (CREATE OBJECT) will result in the automatic execution of the airline method add_airplane.

1-1
In the UML diagram decide what steps are needed where for the triggering and handling of the event airplane_created.
See next page for UML diagram.

1-2
Trigger the event airplane_created in a suitable method within the class lcl_airplane.

1-3
Handle the event within the class lcl_carrier using the handler method add_airplane. This method requires a new interface, its implementation must also be changed slightly.

1-4
In the main program, comment out the calls
r_carrier->add_airplane.

1-5
In the debugger, check whether the event is triggered and handled by the event handler method when the planes are created.
If this does not happen, check whether one of the four important steps for implementing events was perhaps left out.

1-6
 Optional:

Implement the event vehicle_created for the car rental company and its corresponding vehicles.

2
Optional (advanced): 

In the UML, there is the possibility of using events when creating the business partners of the travel agency.
If airlines, car rental companies, or hotels are created, these business partners should automatically be made known to the travel agency.

2-1
You could solve this using the events carrier_created, rental_created, and hotel_created. Would this be problematic?
What would be the best solution?


2-2

Implement your solution for lcl_carrier and lcl_rental.

Defining and triggering events

l

Registering and handling events

Contents:

Events


Solutions

	Unit:
Events
Topic:
Triggering and Handling Events


*&---------------------------------------------------------------------*

*& Report  SAPBC401_EVES_MAIN_A                                        *
*&                                                                     *

*&---------------------------------------------------------------------*

*&       Implement Events in lcl_vehicle and lcl_airplane              *
*&---------------------------------------------------------------------*

REPORT  sapbc401_eves_main_a.
TYPES: ty_fuel TYPE p DECIMALS 2,
       ty_cargo TYPE p DECIMALS 2.
INCLUDE <icon>.
include sapbc401_vehd_h.
INCLUDE sapbc401_eves_a.
DATA: r_vehicle TYPE REF TO lcl_vehicle,
      r_truck TYPE REF TO lcl_truck,
      r_bus   TYPE REF TO lcl_bus,
      r_passenger type ref to lcl_passenger_plane,
      r_cargo type ref to lcl_cargo_plane,
      r_carrier type ref to lcl_carrier,
      r_rental type ref to lcl_rental,
      r_agency type ref to lcl_travel_agency.
START-OF-SELECTION.
*########################

***** Create TRAVEL_AGENCY **************************************
    CREATE OBJECT r_agency EXPORTING im_name = 'Fly&Smile Travel'.
***** Create CARRIER ********************************************
  create object r_carrier exporting im_name = 'Smile&Fly-Travel'.
***** Passenger Plane ********************************************
  CREATE OBJECT r_passenger EXPORTING
                         im_name = 'LH BERLIN'
                         im_planetype = '747-400'
                         im_seats = 345.
***** cargo Plane ************************************************
  CREATE OBJECT r_cargo EXPORTING
                         im_name = 'US HErcules'
                         im_planetype = '747-500'
                         im_cargo = 533.
***** insert planes into itab if client ***************************
*  r_carrier->add_airplane( r_passenger ).
*  r_carrier->add_airplane( r_cargo ).
***** insert business-parnter of agency into partner_list***********
  r_agency->add_partner( r_carrier ).
******* create RENTAL *****************************************
  CREATE OBJECT r_rental EXPORTING im_name = 'HAPPY CAR RENTAL'.
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'MAN'
                                  im_cargo = 45.
*  r_rental->add_vehicle( r_truck ).
******* create truck *****************************************
  CREATE OBJECT r_bus EXPORTING im_make = 'Mercedes'
                                im_passengers = 80.
*  r_rental->add_vehicle( r_bus ).
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'VOLVO'
                                  im_cargo = 48.
*  r_rental->add_vehicle( r_truck ).
***** insert business-parnter of agency into partner_list***********
  r_agency->add_partner( r_rental ).
******* show attributes of all partners of travel_agency ******
  r_agency->display_agency_partners( ).
*&---------------------------------------------------------------------*

*&  Include           SAPBC401_EVES_A                                  *
*&---------------------------------------------------------------------*

*------------------------------------------------------------------*

*       CLASS lcl_airplane DEFINITION                              *
*------------------------------------------------------------------*

CLASS lcl_airplane DEFINITION.
  PUBLIC SECTION.
    "---------------------------------------------

    CONSTANTS: pos_1 TYPE i VALUE 30.
    METHODS: constructor IMPORTING
                   im_name      TYPE string
                   im_planetype TYPE saplane-planetype,
             display_attributes.
    CLASS-METHODS: display_n_o_airplanes.
    events airplane_created.
  PRIVATE SECTION.
    "----------------------------------------------

    METHODS: get_technical_attributes
                 IMPORTING im_type     TYPE saplane-planetype
                 EXPORTING ex_weight   TYPE s_plan_wei
                           ex_tankcap  TYPE s_capacity.
    DATA: name      TYPE string,
          planetype TYPE saplane-planetype.
    CLASS-DATA: n_o_airplanes TYPE i.
ENDCLASS.                    "lcl_airplane DEFINITION
*------------------------------------------------------------------*

*       CLASS lcl_airplane IMPLEMENTATION                          *
*------------------------------------------------------------------*

CLASS lcl_airplane IMPLEMENTATION.
  METHOD constructor.
    name          = im_name.
    planetype     = im_planetype.
    n_o_airplanes = n_o_airplanes + 1.
    raise event airplane_created.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    DATA: weight TYPE saplane-weight,
          cap TYPE saplane-tankcap.
    WRITE: / icon_ws_plane AS ICON,
           / 'Name of airplane'(001), AT pos_1 name,
           / 'Type of airplane: '(002), AT pos_1 planetype.
    get_technical_attributes( EXPORTING im_type = planetype
                              IMPORTING ex_weight = weight
                                        ex_tankcap = cap ).
    WRITE: / 'Weight:'(003), weight,
             'Tankkap:'(004), cap.
  ENDMETHOD.                    "display_attributes
  METHOD display_n_o_airplanes.
    WRITE: /, / 'Number of airplanes: '(ca1),
           AT pos_1 n_o_airplanes LEFT-JUSTIFIED, /.
  ENDMETHOD.                    "display_n_o_airplanes
  METHOD get_technical_attributes.
    SELECT SINGLE weight tankcap FROM saplane
              INTO (ex_weight, ex_tankcap)
              WHERE planetype = im_type.
    IF sy-subrc <> 0.
      ex_weight = 100000.
      ex_tankcap = 10000.
    ENDIF.
  ENDMETHOD.                    "get_technical_attributes
ENDCLASS.                    "lcl_airplane IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_cargo_plane DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_cargo_plane DEFINITION INHERITING FROM lcl_airplane.
  PUBLIC SECTION.
    "----------------------

    METHODS: constructor IMPORTING im_name TYPE string
                                   im_planetype TYPE saplane-planetype
                                   im_cargo TYPE scplane-cargomax.
    METHODS: display_attributes REDEFINITION.
  PRIVATE SECTION.
    "----------------------

    DATA: max_cargo TYPE scplane-cargomax.
ENDCLASS.                    "lcl_cargo_plane DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_cargo_plane IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_cargo_plane IMPLEMENTATION.
  METHOD constructor.
    CALL METHOD super->constructor
      EXPORTING
        im_name      = im_name
        im_planetype = im_planetype.
    max_cargo = im_cargo.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    super->display_attributes( ).
    WRITE: / 'Max Cargo = ', max_cargo.
    ULINE.
  ENDMETHOD.                    "display_attributes
ENDCLASS.                    "lcl_cargo_plane IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_passenger_plane DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_passenger_plane DEFINITION INHERITING FROM lcl_airplane..
  PUBLIC SECTION.
    METHODS: constructor IMPORTING im_name TYPE string
                                   im_planetype TYPE saplane-planetype
                                   im_seats TYPE sflight-seatsmax.
    METHODS: display_attributes REDEFINITION.
  PRIVATE SECTION.
    DATA: max_seats TYPE sflight-seatsmax.
ENDCLASS.                    "lcl_passenger_plane DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_passenger_plane IMPLEMENTATION
*---------------------------------------------------------------------*

*

*---------------------------------------------------------------------*

CLASS lcl_passenger_plane IMPLEMENTATION.
  METHOD constructor.
    CALL METHOD super->constructor
      EXPORTING
        im_name      = im_name
        im_planetype = im_planetype.
    max_seats = im_seats.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    super->display_attributes( ).
    WRITE: / 'Max Seats = ', max_seats.
    ULINE.
  ENDMETHOD.                    "display_attributes
ENDCLASS.                    "lcl_passenger_plane IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_carrier DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_carrier DEFINITION.
  PUBLIC SECTION.
    "----------------------------------------

    INTERFACES lif_partners.
    METHODS: constructor IMPORTING im_name TYPE string,
             get_name RETURNING value(ex_name) TYPE string,
             add_airplane for event airplane_created of lcl_airplane
                                          IMPORTING sender,
             display_airplanes,
             display_attributes.
  PRIVATE SECTION.
    "-----------------------------------

    DATA: name              TYPE string,
          airplane_list TYPE TABLE OF REF TO lcl_airplane.
ENDCLASS.                    "lcl_carrier DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_carrier IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_carrier IMPLEMENTATION.
  METHOD lif_partners~display_partner.
    display_airplanes( ).
  ENDMETHOD.                    "lif_partners~display_partner
  METHOD add_airplane.
    APPEND sender TO airplane_list.
  ENDMETHOD.                    "add_airplane
  METHOD display_attributes.
    WRITE: icon_flight AS ICON, name . ULINE. ULINE.
    display_airplanes( ).
  ENDMETHOD.                    "display_attributes
  METHOD display_airplanes.
    DATA: r_plane TYPE REF TO lcl_airplane.
    LOOP AT airplane_list INTO r_plane.
      r_plane->display_attributes( ).
    ENDLOOP.
  ENDMETHOD.                    "display_airplanes
  METHOD constructor.
    name = im_name.
    set handler add_airplane for all instances.
ENDMETHOD.                    "constructor
  METHOD get_name.
    ex_name = name.
  ENDMETHOD.                    "get_name
ENDCLASS.                    "lcl_carrier IMPLEMENTATION
Solutions (optional)

*&---------------------------------------------------------------------*

*& Report  SAPBC401_EVES_MAIN_B                                        *
*&                                                                     *

*&---------------------------------------------------------------------*

*&       Implement Events in LCL_vehicle and lcl_airplane              *
*&       Implement Events in LCL_carrier and lcl_rental                *
*&---------------------------------------------------------------------*

REPORT  sapbc401_eves_main_b.
TYPES: ty_fuel TYPE p DECIMALS 2,
       ty_cargo TYPE p DECIMALS 2.
INCLUDE <icon>.
include sapbc401_vehd_i.
INCLUDE sapbc401_eves_b.
DATA: r_vehicle TYPE REF TO lcl_vehicle,
      r_truck TYPE REF TO lcl_truck,
      r_bus   TYPE REF TO lcl_bus,
      r_passenger type ref to lcl_passenger_plane,
      r_cargo type ref to lcl_cargo_plane,
      r_carrier type ref to lcl_carrier,
      r_rental type ref to lcl_rental,
      r_agency type ref to lcl_travel_agency.
START-OF-SELECTION.
*########################

******* create travel_agency *****************************************
  CREATE OBJECT r_agency EXPORTING im_name = 'Fly&Smile Travel'.
******* create rental *****************************************
  CREATE OBJECT r_rental EXPORTING im_name = 'HAPPY CAR RENTAL'.
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'MAN'
                                  im_cargo = 45.
******* create truck *****************************************
  CREATE OBJECT r_bus EXPORTING im_make = 'Mercedes'
                                im_passengers = 80.
******* create truck *****************************************
  CREATE OBJECT r_truck EXPORTING im_make = 'VOLVO'
                                  im_cargo = 48.
***** Create CARRIER ********************************************
  create object r_carrier exporting im_name = 'Smile&Fly-Travel'.
***** Passenger Plane ********************************************
  CREATE OBJECT r_passenger EXPORTING
                         im_name = 'LH BERLIN'
                         im_planetype = '747-400'
                         im_seats = 345.
***** cargo Plane ************************************************
  CREATE OBJECT r_cargo EXPORTING
                         im_name = 'US HErcules'
                         im_planetype = '747-500'
                         im_cargo = 533.
******* show attributes of all partners of travel_agency ******
  r_agency->display_agency_partners( ).
*&---------------------------------------------------------------------*

*&  Include           SAPBC401_EVES_B                                  *
*&---------------------------------------------------------------------*

*------------------------------------------------------------------*

* events in: lcl_airplane and lcl_carrier !
*

*------------------------------------------------------------------*

*       CLASS lcl_airplane DEFINITION                              *
*------------------------------------------------------------------*

CLASS lcl_airplane DEFINITION.
  PUBLIC SECTION.
    "---------------------------------------------

    CONSTANTS: pos_1 TYPE i VALUE 30.
    METHODS: constructor IMPORTING
                   im_name      TYPE string
                   im_planetype TYPE saplane-planetype,
             display_attributes.
    CLASS-METHODS: display_n_o_airplanes.
    EVENTS:airplane_created.
  PRIVATE SECTION.
    "----------------------------------------------

    METHODS: get_technical_attributes
                 IMPORTING im_type     TYPE saplane-planetype
                 EXPORTING ex_weight   TYPE s_plan_wei
                           ex_tankcap  TYPE s_capacity.
    DATA: name      TYPE string,
          planetype TYPE saplane-planetype.
    CLASS-DATA: n_o_airplanes TYPE i.
ENDCLASS.                    "lcl_airplane DEFINITION
*------------------------------------------------------------------*

*       CLASS lcl_airplane IMPLEMENTATION                          *
*------------------------------------------------------------------*

CLASS lcl_airplane IMPLEMENTATION.
  METHOD constructor.
    name          = im_name.
    planetype     = im_planetype.
    n_o_airplanes = n_o_airplanes + 1.
    RAISE EVENT airplane_created.
  ENDMETHOD.                    "constructor
  METHOD display_attributes.
    DATA: weight TYPE saplane-weight,
          cap TYPE saplane-tankcap.
    WRITE: / icon_ws_plane AS ICON,
           / 'Name of airplane'(001), AT pos_1 name,
           / 'Type of airplane: '(002), AT pos_1 planetype.
    get_technical_attributes( EXPORTING im_type = planetype
                              IMPORTING ex_weight = weight
                                        ex_tankcap = cap ).
    WRITE: / 'Weight:'(003), weight,
             'Tankkap:'(004), cap.
  ENDMETHOD.                    "display_attributes
  METHOD display_n_o_airplanes.
    WRITE: /, / 'Number of airplanes: '(ca1),
           AT pos_1 n_o_airplanes LEFT-JUSTIFIED, /.
  ENDMETHOD.                    "display_n_o_airplanes
  METHOD get_technical_attributes.
    SELECT SINGLE weight tankcap FROM saplane
              INTO (ex_weight, ex_tankcap)
              WHERE planetype = im_type.
    IF sy-subrc <> 0.  ex_weight = 100000. ex_tankcap = 10000. ENDIF.
  ENDMETHOD.                    "get_technical_attributes
ENDCLASS.                    "lcl_airplane IMPLEMENTATION
...

*---------------------------------------------------------------------*

*       CLASS lcl_carrier DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_carrier DEFINITION.
  PUBLIC SECTION.
    "----------------------------------------

    INTERFACES lif_partners.
    METHODS: constructor IMPORTING im_name TYPE string,
             get_name RETURNING value(ex_name) TYPE string,
             add_airplane FOR EVENT airplane_created OF lcl_airplane
                       IMPORTING sender,
  PRIVATE SECTION.
    "-----------------------------------

    DATA: name              TYPE string,
          airplane_list TYPE TABLE OF REF TO lcl_airplane.
ENDCLASS.                    "lcl_carrier DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_carrier IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_carrier IMPLEMENTATION.
  METHOD lif_partners~display_partner.
    display_attributes( ).
  ENDMETHOD.                    "lif_partners~display_partner
  METHOD add_airplane.
    APPEND sender TO airplane_list.
  ENDMETHOD.                    "add_airplane
  METHOD display_attributes.
    skip 2.
    WRITE: icon_flight AS ICON, name . ULINE. ULINE.
    display_airplanes( ).
  ENDMETHOD.                    "display_attributes
  METHOD display_airplanes.
    DATA: r_plane TYPE REF TO lcl_airplane.
    LOOP AT airplane_list INTO r_plane.
      r_plane->display_attributes( ).
    ENDLOOP.
  ENDMETHOD.                    "display_airplanes
  METHOD constructor.
    name = im_name.
    SET HANDLER add_airplane FOR ALL INSTANCES.
    RAISE EVENT lif_partners~partner_created.
  ENDMETHOD.                    "constructor
ENDCLASS.                    "lcl_carrier IMPLEMENTATION
*&---------------------------------------------------------------------*

*&  Include           SAPBC401_VEHD_i                                  *
*&---------------------------------------------------------------------*

*---------------------------------------------------------------------*

* define client lcl_travel_agency
* it will use the interface lif_partners
*

* implement EVENT in LCL_VEHICLE and LCL_RENTAL
*---------------------------------------------------------------------*

INTERFACE lif_partners.
  METHODS display_partner.
*** event defined inside the interface !! ****

  EVENTS: partner_created.
ENDINTERFACE.                    "lif_partners
*---------------------------------------------------------------------*

*       CLASS lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_vehicle DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS: get_average_fuel IMPORTING im_distance TYPE s_distance
                                        im_fuel TYPE ty_fuel
                              RETURNING value(re_avgfuel) TYPE ty_fuel.
    METHODS       constructor IMPORTING im_make TYPE string.
    METHODS       display_attributes.
    METHODS       set_make IMPORTING im_make TYPE string.
    METHODS       get_make EXPORTING ex_make TYPE string.
    CLASS-METHODS get_count EXPORTING re_count TYPE i.
    EVENTS: vehicle_created.
  PRIVATE SECTION.
    "-------------------

    DATA: make   TYPE string.
    METHODS      init_make.
    CLASS-DATA:  n_o_vehicles TYPE i.
ENDCLASS.                    "lcl_vehicle DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_vehicle IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_vehicle IMPLEMENTATION.
  METHOD get_average_fuel.
    re_avgfuel = im_distance / im_fuel.
  ENDMETHOD.                    "get_average_fuel
  METHOD constructor.
    make = im_make.
    ADD 1 TO n_o_vehicles.
    raise event vehicle_created.
  ENDMETHOD.                    "constructor
  METHOD set_make.
    IF im_make IS INITIAL.
      init_make( ).   " me->init_make( ). also possible
    ELSE.
      make = im_make.
    ENDIF.
  ENDMETHOD.                    "set_make
  METHOD init_make.
    make = 'default make'.
  ENDMETHOD.                    "init_make
  METHOD get_make.
    ex_make = make.
  ENDMETHOD.                    "get_make
  METHOD display_attributes.
    WRITE: make.
  ENDMETHOD.                    "display_attributes
  METHOD get_count.
    re_count = n_o_vehicles.
  ENDMETHOD.                    "get_count
ENDCLASS.                    "lcl_vehicle IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_rental DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_rental DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS:      constructor IMPORTING im_name TYPE string.
    METHODS       add_vehicle for event vehicle_created of lcl_vehicle
                         importing sender.
    METHODS       display_attributes.
    INTERFACES: lif_partners.
  PRIVATE SECTION.
    "-------------------

    DATA: name TYPE string,
          vehicle_list TYPE TABLE OF REF TO lcl_vehicle.
ENDCLASS.                    "lcl_rental DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_rental IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_rental IMPLEMENTATION.
  METHOD lif_partners~display_partner.
    display_attributes( ).
  ENDMETHOD.                    "lif_partners~display_partner
  METHOD  constructor.
    name = im_name.
    set handler add_vehicle for all instances.
    raise event lif_partners~partner_created.
  ENDMETHOD.                    "constructor
  METHOD  add_vehicle.
    APPEND sender TO vehicle_list.
  ENDMETHOD.                    "add_vehicle
  METHOD  display_attributes.
    DATA: r_vehicle TYPE REF TO lcl_vehicle.
    skip 2.
    WRITE: /  icon_transport_proposal AS ICON, name.
    WRITE:  ' Here comes the vehicle list: '. ULINE. ULINE.
    LOOP AT vehicle_list INTO r_vehicle.
      r_vehicle->display_attributes( ).
    ENDLOOP.
  ENDMETHOD.                    "display_attributes
ENDCLASS.                    "lcl_rental IMPLEMENTATION
*---------------------------------------------------------------------*

*       CLASS lcl_travel_agency DEFINITION
*---------------------------------------------------------------------*

CLASS lcl_travel_agency DEFINITION.
  PUBLIC SECTION.
    "-------------------

    METHODS:   constructor IMPORTING im_name TYPE string.
    METHODS    add_partner for event partner_created of lif_partners
                               IMPORTING sender.
    METHODS       display_agency_partners.
  PRIVATE SECTION.
    "-------------------

    DATA: name TYPE string,
          partner_list TYPE TABLE OF REF TO lif_partners.
ENDCLASS.                    "lcl_travel_agency DEFINITION
*---------------------------------------------------------------------*

*       CLASS lcl_travel_agency IMPLEMENTATION
*---------------------------------------------------------------------*

CLASS lcl_travel_agency IMPLEMENTATION.
  METHOD display_agency_partners.
    DATA: r_partner TYPE REF TO lif_partners.
    WRITE: icon_dependents AS ICON, name.
    WRITE:  ' Here are the partners of the travel agency: '.ULINE.ULINE.
    LOOP AT partner_list INTO r_partner.
      r_partner->display_partner( ).
    ENDLOOP.
ENDMETHOD.                    "display_agency_partners
METHOD  constructor.
  name = im_name.
  set handler add_partner for all instances.
ENDMETHOD.                    "constructor
METHOD  add_partner.
  APPEND sender TO partner_list.
ENDMETHOD.                    "add_partner
ENDCLASS.                    "lcl_travel_agency IMPLEMENTATION

